Texture,
1974, Vol. 1, pp. 195–199

AUTOMATIC DRAWING OF STEREOGRAPHIC NETS

A. VADON, D. RUER and R. BÄRO

Laboratoire de Métallurgie Structurale, Faculté des Sciences, Université de Metz, France

(*Received January 10, 1973*)

The present paper describes a computer program for the automatic tracing of new stereographic nets aimed at the description of pole figures.

INTRODUCTION

The new, recently published stereo nets were designed to facilitate the description of pole figures.[1] Their use has allowed us to develop a method based on the partial-fibre concept for analysing qualitatively rolling textures.[2]

These nets come in a standard execution: the radius is 10 cm, the angle ω of the pole's sphere axis with the projection plane is a multiple of 5°, and the minimal angle between two parallels or two meridians is 10°.

For some studies, it may be useful to choose nets with different diameters, corresponding to angles ω which are not multiples of 5°. Besides these, it is sometimes desirable to have more precise nets, capable for measuring angular distances smaller than 10° without interpolation.

The construction by hand of a net based on the published equations is very tedious. Programs for automatic tracing of stereographic projections and the Lambert projection have been published respectively by Starkey[4] and by Wenk and Trommsdorff.[5] However, the method used by these authors is based on the resolution of spherical triangles which for the use of the stereographic projections involves great complications and leads to programs of considerable length.

To overcome these difficulties, we have perfected a new program for automatic tracing, which produces quickly stereographic nets of any desired specifications.

PRINCIPLE OF OPERATION

The fundamental operation consists of the following:

a) the rotation by an angle α of a point M on the reference sphere (Σ) about an axis $\mathbf{\Omega'\Omega}$ (Figure 1) transforms it into M_1.

b) the stereographic projections of M and M_1 (and of the arc MM_1) are, respectively M' and M'_1 (and the arc $M'M'_1$).

The angles $\theta = (\mathbf{O}_z, \mathbf{O\Omega}) = \pi/2 - \omega$ and $\psi = (\mathbf{O}_x, \mathbf{OO'})$ define the position of the axis $\mathbf{\Omega'\Omega}$. The arc η represents the colatitude of points M and M_1 (measured on the sphere of axis $\mathbf{\Omega\Omega'}$).

The coordinates of M can be expressed in the form of:

$$\begin{aligned} X_M &= R\sin(\theta-\eta)\cos\psi \\ Y_M &= R\sin(\theta-\eta)\sin\psi \\ Z_M &= R\cos(\theta-\eta) \end{aligned} \tag{1}$$

The coordinates of M_1, transformed from M by the rotation ($\mathbf{O\Omega}$, α), are obtained by:

$$\begin{bmatrix} X_{M1} \\ Y_{M_1} \\ Z_{M_1} \end{bmatrix} = A \begin{bmatrix} X_M \\ Y_M \\ Z_M \end{bmatrix} \tag{2}$$

where A is a matrix of rotation which can be decomposed into a sum of three matrices containing the terms of the angle α and the components C_1, C_2, C_3 of the axis $\mathbf{O\Omega}$.[3] Matrix A is thus written as:

$$\begin{aligned} A = \cos\alpha \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \\ +(1-\cos\alpha) \begin{bmatrix} C_1^2 & C_1C_2 & C_1C_3 \\ C_2C_1 & C_2^2 & C_2C_3 \\ C_3C_1 & C_3C_2 & C_3^2 \end{bmatrix} \\ +\sin\alpha \begin{bmatrix} C & -C_3 & C_2 \\ C_3 & 1 & C_1 \\ -C_2 & C_1 & 0 \end{bmatrix} \end{aligned} \tag{3}$$

To obtain all the M_1 points located on the parallel passing through M one can just let α vary from 0 to 2π (See in annex: subroutine ROTAS).

The stereographic projection M' of point M on the $x0y$ plane can be obtained by expressing that vectors $\mathbf{PM'}$ and $\mathbf{M'M}$ are colinear (Fig. 1)

$$X_{M'} = \frac{ZpX_M}{Zp-Z_M}$$
$$Y_{M'} = \frac{ZpY_M}{Zp-Z_M} \tag{4}$$



FIGURE 1 Stereographic projection of a pole describing the parallel of latitude $(\pi/2-\eta)$ on the sphere of axis $\Omega'\Omega$. OE is the axis of the meridian on which M is located. α is the difference in longitude of M_1 and M_1'.

where $Zp = \mp R$, depending upon whether the height of M is positive (as in Fig. 1) or negative.

The stereographic projection of the parallels can now be obtained by making η to vary.

For the particular value of $\eta = \pi/2$ relation (2) gives the projections of the vectors **OE** which are perpendicular to **OΩ** (Figure 1). Each of these vectors is the axis of one meridian of diameter **Ω′Ω** and longitude α. So the coordinates of any point M_1 of a meridian are given by:

$$\begin{bmatrix} X_{M_1} \\ Y_{M_1} \\ Z_{M_1} \end{bmatrix} = A' \begin{bmatrix} X_\Omega \\ Y_\Omega \\ Z_\Omega \end{bmatrix} \tag{5}$$

The matrix of relation (5) can be obtained when one replaces in the matrix of relation (3) the components of **OΩ** by the components of **OE** which is perpendicular to plane ($\mathbf{O\Omega}, \mathbf{OM_1}$). The stereographic projection of the meridians can thus be derived directly from relation (5).

Figure 2 shows a net for which $\omega = \pi/2-\theta = 32.5°$ and $\psi = 0°$, with the parallels and meridians separated by angles of 10°. The stereographic projection of the hemisphere under the projection plane is drawn by dots, the distance between two dots corresponding to 1°. The drawing has been done automatically by a BENSON 120 plotter conducted by an IBM 1130 computer which was programmed in FORTRAN IV. The absolute error in the drawing of a net is 0.1 mm. The net shown in Figure 2 has been obtained in about 30 minutes. The radius R of the net can be chosen at will up to 35 cm. The ordinogram is given in Figure 3 and the program in the Appendix.

ACKNOWLEDGMENT

We express our sincere gratitude to Monsieur P. Maurice, Director of the Ecole Nationale d'Ingénieurs de Metz for allowing us to use the computer and the plotter of this Institute.



FIGURE 2 Stereographic net corresponding to $\omega = 32.5°$. On the dotted lines the angular distance between two dots is 1°.



FIGURE 3 Ordinogram.

REFERENCES

1. D. Ruer and R. Baro, *J. Appl. Cryst.* **4**, 321 (1971).
2. D. Ruer, A. Vadon and R. Baro, to be published.
3. G. A. Korn and T. M. Korn, *Mathematical Handbook for Scientists and Engineers* (McGraw Hill Book Company, New York, 1968) p. 473.
4. J. Starkey, in *Experimental and Natural Rock Deformation*, P. Paulitsch, Ed. (Springer Verlag, Berlin/Heidelberg/New York, 1970) pp. 38–50.
5. H. R. Wenk und V. Trommsdorff, Koordinatentransformation, Mittelbare Orientierung, Nachbarwinkelstatistik. Beiträge zur Mineralogie und Petrographie **11**, 559 (1965).

APPENDIX

Program of drawing

```
C
C
C
C
C
C
C
C
C
C
C
C
C
C
C
C
C
C         CHOICE OF UNITS
C
      READ(2,1)UX
    1 FORMAT(F10.5)
      CALL SCALF(UX,UX,0.,0.)
      PI=3.1415926
      P=PI/180.
      R=1.
C         THE RADIUS OF THE REFERENCE SPHERE IS TAKEN FOR UNIT
C
C         DRAWING OF PROJECTION CIRCLE
C
      CALL CERCL(0.,0.,1.,UX)
C
C         READING OF INPUTS
C
      READ(2,2)OMEGA,PSI
    2 FORMAT(2F10.5)
      WRITE(5,3)OMEGA,PSI
    3 FORMAT(/10X,'OMEGA=',F5.2,5X,'PSI=',F5.2)
      TETA=90.-OMEGA
      TETA=TETA*P
      PSI=PSI*P
C
C         COMPONENTS OF THE AXIS OF ROTATION
C
      SP=SIN(PSI)
      CP=COS(PSI)
      XC=SIN(TETA)*CP
      YC=SIN(TETA)*SP
      ZC=COS(TETA)



      DO 100 M=1,2
      DO 100 I=1,180,10
      DO 100 J=1,361
      IF(M-1)4,4,5
C
C         INITIALISATION AND COMPONENTS OF THE AXIS OF THE PARALLELS
C
    4 X=SIN(TETA-FLOAT(I-1)*P)*CP
      Y=SIN(TETA-FLOAT(I-1)*P)*SP
      Z=COS(TETA-FLOAT(I-1)*P)
      C1=XC
      C2=YC
      C3=ZC
      GO TO 6
C
C         INITIALISATION AND COMPONENTS OF THE AXIS OF THE MERIDIANS
C
    5 X=XC
      Y=YC
      Z=ZC
      XE=SP
      YE=-CP
      ZE=0.
      CALL ROTAS(XC,YC,ZC,FLOAT(I-1)*P,XE,YE,ZE)
      C1=XE
      C2=YE
      C3=ZE
C
C         ROTATION,STEREOGRAPHIC PROJECTION AND DRAWING
C
    6 CALL ROTAS(C1,C2,C3,FLOAT(J-1)*P,X,Y,Z)
      IF(Z)7,8,9
    7 XPSA=X*R/(R-Z)
      YPSA=Y*R/(R-Z)
      CALL FPLOT(1,XPSA,YPSA)
      CALL FPLOT(-2,XPSA,YPSA)
      GO TO 11
    8 XPSA=X
      YPSA=Y
      GO TO 10
    9 XPSA=X*R/(R+Z)
      YPSA=Y*R/(R+Z)
   10 CALL FPLOT(-2,XPSA,YPSA)
   11 IF(J-361)100,12,12
   12 CALL FPLOT(+1,XPSA,YPSA)
      IF(I-171)100,13,13
C
C         I MINUS THE LAST VALUE OF I
C
   13 PAUSE
  100 CONTINUE
      CALL EXIT
      END

FEATURES SUPPORTED
 ONE WORD INTEGERS
 IOCS

CORE REQUIREMENTS FOR VT2
 COMMON        0  VARIABLES      54  PROGRAM      480
```

```
      SUBROUTINE CERCL(XC,YC,R,UX)
C
C           THIS SUBROUTINE ALLOWS THE DRAWING OF A CIRCLE HAVING
C           * (XC,YC) FOR CENTER
C           *. R FOR RADIUS EXPRESSED IN UNITS OF THE SCALFROUTINE
C           * THE UNIT ON AXES OX AND OY HAS THE SAME VALUE UX EXPESSED
C         IN CM
C         UX IS DETERMINED BY THE SCALFROUTINE
C
C         THE DRAWING INCREMENT IS 0.1 CM
C
      PI=3.1415926
      CALL FPLOT(-2, XC+R,YC)
      N=20.*PI*R*UX+1.
      DO 1 I=1,N
      AI=I .
      PAS=0.1/(R*UX)
      T=AI*PAS
      X=R*COS(T)
      Y=R*SIN(T)
    1 CALL FPLOT(0,XC+X,YC+Y)
      CALL FPLOT(-1,XC+R,YC)
      RETURN
      END

FEATURES SUPPORTED
 ONE WORD INTEGERS

CORE REQUIREMENTS FOR CERCL
 COMMON        0  VARIABLES      20  PROGRAM     150
```

```
      SUBROUTINE ROTAS(C1,C2,C3,ALPHA,X,Y,Z)
C
C         THIS SUBROUTINE ALLOWS TO CALCULATE THE COORDINATES OF M1,
C       TRANSFORMED FROM M(X,Y,Z) BY A ROTATION HAVING
C           * FOR AXIS THE COMPONENTS C1,C2,C3
C           * FOR ANGLE ALPHA IN RADIANS
C
C         AFTER THE ROTATION THE COORDINATES ARE CALLED X,Y,Z
C
      SB=SIN(ALPHA)
      CB=COS(ALPHA)
      S1=C1*SB
      S2=C2*SB
      S3=C3*SB
      CCB=1.-CB
      C12=C1*C2*CCB
      C13=C1*C3*CCB
      C23=C2*C3*CCB
C
      XA=(CB+C1*C1*CCB)*X+(C12-S3)*Y+(C13+S2)*Z
      YA=(C12+S3)*X+(CB+C2*C2*CCB)*Y+(C23-S1)*Z
      ZA=(C13-S2)*X+(C23+S1)*Y+(CB+C3*C3*CCB)*Z
      X=XA
      Y=YA
      Z=ZA
      RETURN
      END

FEATURES SUPPORTED
 ONE WORD INTEGERS

CORE REQUIREMENTS FOR ROTAS
 COMMON        0  VARIABLES      28  PROGRAM     210
```